KAIST
School of Computing

Web3@KAIST
Building Web3 Apps to Solve Real Problems

Building Web3 & Blockchain Applications
(CS492 Special Topics in Computer Science)
Spring 2023

Developing Governance

Lecture 19 (2023-05-17)

Jason Han, Ph.D

Adjunct Professor of KAIST School of Computing
Founder of Ground X & Klaytn

web3classdao@gmail.com
http://web3classdao.xyz/kaist/

Today's Lecture 17 Overview

- **Lecture Objective**
  - Understanding access control in a smart contract
  - Understanding on-chain and off-chain governance
  - Learning how to develop on-chain governance in Solidity
  - Learning how to upgrade a smart contract

- **Lecture will cover**
  - Access Control
  - Governance
  - Upgradeable contracts

References for the lecture

- Dapp University: Governance (Github)
- Ultimate Web3, Full Stack Solidity, and Smart Contract Course by Patrick Collins

  - Lesson 17: Hardhat DAOs

  - Lesson 16: Hardhat Upgrades
- OpenZeppelin access control
- OpenZeppelin governance
- OpenZeppelin governance code
- How to Create a DAO Governance Token (Alchemy)
- Compound governance
- Tally: Explore DAOs
- Open Zeppelin Upgradeable Contracts (Github)
- Open Zeppelin Upgradeable ERC20
- Writing Upgradeable Contracts (OpenZeppelin)
- OpenZeppelin Upgrades: Step by Step Tutorial for Hardhat

A governance & upgradeable contracts

Examples from various sites
with some modification

Clone the code here!

git clone https://github.com/web3classdao/onchain-governance.git
git clone https://github.com/web3classdao/upgradeable-contracts.git

Access Control

Access Control so far

- *The contract deployer (owner) by EOA*
- *The contract deployer (owner) by another contract*
- *msg.sender (usually EOA/contracts calling a function)*

Need more granular access control

OpenZeppelin Ownable.sol

Use the modifier to check the calling address is the owner



```
8  abstract contract Ownable is Context {
9      address private _owner;
10
11     event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);
12
13     constructor() {
14         _transferOwnership(_msgSender());
15     }
16
17     modifier onlyOwner() {
18         _checkOwner();
19         _;
20     }
21
22     function _checkOwner() internal view virtual {
23         require(owner() == _msgSender(), "Ownable: caller is not the owner");
24     }
25
26     function transferOwnership(address newOwner) public virtual onlyOwner {
27         require(newOwner != address(0), "Ownable: new owner is the zero address");
28         _transferOwnership(newOwner);
29     }
```

OpenZeppelin AccessControl.sol

addresses that have a role
the admin of a role

Use the modifier
to check the calling account has the role

Granting and revoking a role are
the same as changing the _roles variable

```
abstract contract AccessControl is Context, IAccessControl, ERC165 {
    struct RoleData {
        mapping(address => bool) members;
        bytes32 adminRole;
    }

    mapping(bytes32 => RoleData) private _roles;

    bytes32 public constant DEFAULT_ADMIN_ROLE = 0x00;

    modifier onlyRole(bytes32 role) {
        _checkRole(role);
        _;
    }

    function _checkRole(bytes32 role) internal view virtual {
        _checkRole(role, _msgSender());
    }

    function _checkRole(bytes32 role, address account) internal view virtual {
        if (!hasRole(role, account)) {
            revert(
                string(
                    abi.encodePacked(
                        "AccessControl: account ",
                        Strings.toHexString(account),
                        " is missing role ",
                        Strings.toHexString(uint256(role), 32)
                    )
                )
            );
        }
    }

    function hasRole(bytes32 role, address account)
        public view virtual override returns (bool) {
        return _roles[role].members[account];
    }

    function _grantRole(bytes32 role, address account) internal virtual {
        if (!hasRole(role, account)) {
            _roles[role].members[account] = true;
            emit RoleGranted(role, account, _msgSender());
        }
    }

    function _revokeRole(bytes32 role, address account) internal virtual {
        if (hasRole(role, account)) {
            _roles[role].members[account] = false;
            emit RoleRevoked(role, account, _msgSender());
        }
    }
```

Role-based ERC20 Token Example

A role is set as bytes32 constant to save the storage

Use _grantRole internal function of AccessControl

Use the onlyRole modifier with a role argument

RoleBasedToken.sol

```sol
// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/access/AccessControl.sol";
import "@openzeppelin/contracts/token/ERC20/ERC20.sol";

contract RoleBasedToken is ERC20, Ownable, AccessControl {
    bytes32 public constant MINTER_ROLE = keccak256("MINTER_ROLE");
    bytes32 public constant BURNER_ROLE = keccak256("BURNER_ROLE");

    constructor(address minter, address burner) ERC20("MyTestToken", "MTT") {
        _grantRole(MINTER_ROLE, minter);
        _grantRole(BURNER_ROLE, burner);
    }

    function grantRole(bytes32 role, address account) public override onlyOwner {
        _grantRole(role, account);
    }

    function mint(address to, uint256 amount) public onlyRole(MINTER_ROLE) {
        _mint(to, amount);
    }

    function burn(address from, uint256 amount) public onlyRole(BURNER_ROLE) {
        _burn(from, amount);
    }
}
```

Role-based ERC20 Token Example

A role is set as bytes32 constant to save the storage

Use _grantRole internal function of AccessControl

Grant the deployer the admin role If granting roles more accounts, the deployer can call the grantRole()

Use the onlyRole modifier with a role argument

RoleBasedToken.sol

```
1   // SPDX-License-Identifier: MIT
2   pragma solidity ^0.8.0;
3
4   import "@openzeppelin/contracts/access/AccessControl.sol";
5   import "@openzeppelin/contracts/token/ERC20/ERC20.sol";
6
7   contract RoleBasedToken is ERC20, AccessControl {
8       bytes32 public constant MINTER_ROLE = keccak256("MINTER_ROLE");
9       bytes32 public constant BURNER_ROLE = keccak256("BURNER_ROLE");
10
11      constructor(address minter, address burner) ERC20("MyTestToken", "MTT") {
12          _grantRole(MINTER_ROLE, minter);
13          _grantRole(BURNER_ROLE, burner);
14
15          // Grant the contract deployer the default admin role:
16          // it will be able to grant and revoke any roles
17          _grantRole(DEFAULT_ADMIN_ROLE, msg.sender);
18      }
19
20      function mint(address to, uint256 amount) public onlyRole(MINTER_ROLE) {
21          _mint(to, amount);
22      }
23
24      function burn(address from, uint256 amount) public onlyRole(BURNER_ROLE) {
25          _burn(from, amount);
26      }
27  }
```

Governance

What is **Governance?**

Governance

- **Governance is the systems in place that allow decisions to be made**
  - In a typical organizational structure, the executive team or a board of directors may have the final say in decision-making, or shareholders vote on proposals to enact change
  - In a political system, elected officials may enact legislation that attempts to represent their constituents' desires

- **Decentralized governance**: No one person or authority owns or controls the governance
  - Blockchain is used to implement decentralized governance
  - **On-chain governance**: when proposed protocol changes are decided by a stakeholder vote, usually by holders of a governance token, and voting happens on the blockchain. Examples are Compound and Uniswap.
  - **Off-chain governance**: where any protocol change decisions happen through an informal process of social discussion, which, if approved, would be implemented in code. An example is Ethereum.

https://ethereum.org/en/governance/

On-chain vs. Off-chain governance

An on-chain flow runs everything entirely in code:



An off-chain flow depends on trusting admins to execute the result of the vote:



https://docs.tally.xyz/knowledge-base/tally/on-chain-vs-off-chain

On-chain Governance Example

Compound
Service fee payment

https://compound.finance/governance/proposals/157



On-chain Governance Example

Nouns DAO
Governance parameter update

https://www.tally.xyz/gov/nounsdao/proposal/261



On-chain governance lifecycle



On-chain governance example

Proposal: the ETH funds of treasury will be released to the recipient



https://github.com/dappuniversity/governance

On-chain governance architecture

You have to implement 4 contracts:
3 of them just inherit OpenZeppelin governance contracts (red boxes)



https://github.com/dappuniversity/governance

OpenZeppelin Wizard

The easiest way to get template codes for tokens and governance

https://docs.openzeppelin.com/contracts/4.x/wizard



Governance Token inherited by ERC20Votes

*The voting power of each account is **retrieved from past snapshots** rather than current balance, which is an important protection that **prevents double voting***

```
// SPDX-License-Identifier: MIT
pragma solidity ^0.8.9;

import "@openzeppelin/contracts/token/ERC20/extensions/ERC20Votes.sol";

contract Token is ERC20Votes {
    constructor(
        string memory _name,
        string memory _symbol,
        uint256 _initialSupply
    ) ERC20(_name, _symbol) ERC20Permit(_name) {
        _mint(msg.sender, _initialSupply);
    }

    // The functions below are overrides required by Solidity.
    function _afterTokenTransfer(
        address from,
        address to,
        uint256 amount
    ) internal override(ERC20Votes) {
        super._afterTokenTransfer(from, to, amount);
    }

    function _mint(address to, uint256 amount) internal override(ERC20Votes) {
        super._mint(to, amount);
    }

    function _burn(address account, uint256 amount) internal override(ERC20Votes) {
        super._burn(account, amount);
    }
}
```

Timelock inherited by TimelockController

Defining the role of proposers and executors and setting up minDelay

```
1  // SPDX-License-Identifier: MIT
2  pragma solidity ^0.8.9;
3
4  import "@openzeppelin/contracts/governance/TimelockController.sol";
5
6  contract TimeLock is TimelockController {
7    // minDelay is how long you have to wait before executing
8    // proposers is the list of addresses that can propose
9    // executors is the list of addresses that can execute
10   /**
11    * IMPORTANT: The optional admin can aid with initial configuration of roles after deployment
12    * without being subject to delay, but this role should be subsequently renounced in favor of
13    * administration through timelocked proposals. Previous versions of this contract would assign
14    * this admin to the deployer automatically and should be renounced as well.
15    */
16   constructor(
17     uint256 minDelay,
18     address[] memory proposers,
19     address[] memory executors
20   ) TimelockController(minDelay, proposers, executors) {}
21 }
```

How long should a proposal be delayed to be executed after it passes

Who can propose?

Who can execute a proposal?

Governance inherited by Governor

what options people have when casting a vote and how those votes are counted

GovernorCountingSimple
For, Against, and Abstain only For and Abstain votes are counted towards quorum

how voting power is determined

how many votes are needed for quorum

set up roles and a timelock for execution

```
pragma solidity ^0.8.9;

import "@openzeppelin/contracts/governance/Governor.sol";
import "@openzeppelin/contracts/governance/extensions/GovernorCountingSimple.sol";
import "@openzeppelin/contracts/governance/extensions/GovernorVotes.sol";
import "@openzeppelin/contracts/governance/extensions/GovernorVotesQuorumFraction.sol";
import "@openzeppelin/contracts/governance/extensions/GovernorTimelockControl.sol";
import "@openzeppelin/contracts/governance/extensions/GovernorSettings.sol";

contract Governance is
    Governor,
    GovernorCountingSimple,
    GovernorVotes,
    GovernorVotesQuorumFraction,
    GovernorTimelockControl
{
    uint256 public votingDelay_;
    uint256 public votingPeriod_;

    constructor(
        ERC20Votes _token,
        TimelockController _timelock,
        uint256 _quorum,
        uint256 _votingDelay,
        uint256 _votingPeriod
    )
        Governor("Web3@KAIST DAO")
        GovernorVotes(_token)
        GovernorVotesQuorumFraction(_quorum)
        GovernorTimelockControl(_timelock)
    {
        votingDelay_ = _votingDelay;
        votingPeriod_ = _votingPeriod;
    }
```

Treasury contract

ETH fund treasury

- *The fund will be set when the contract is created*
- *Only the owner of the contract can release the fund*

```solidity
// SPDX-License-Identifier: MIT
pragma solidity ^0.8.9;

import "@openzeppelin/contracts/access/Ownable.sol";

contract Treasury is Ownable {
    uint256 public totalFunds;
    address public payee;
    bool public isReleased;

    constructor(address _payee) payable {
        totalFunds = msg.value;
        payee = _payee;
        isReleased = false;
    }

    function releaseFunds() public onlyOwner {
        isReleased = true;
        payable(payee).transfer(totalFunds);
    }
}
```

Deployment Script (1/2)

2_deploy_contracts.js

```
const Token = artifacts.require("Token")
const Timelock = artifacts.require("Timelock")
const Governance = artifacts.require("Governance")
const Treasury = artifacts.require("Treasury")

module.exports = async function (deployer) {

    const [executor, proposer, voter1, voter2, voter3, voter4, voter5] = await web3.eth.getAccounts()

    const name = "Web3@KAIST"
    const symbol = "W3K"
    const supply = web3.utils.toWei('1000', 'ether') // 1000 Tokens

    // Deploy token
    await deployer.deploy(Token, name, symbol, supply)
    const token = await Token.deployed()

    const amount = web3.utils.toWei('50', 'ether')
    await token.transfer(voter1, amount, { from: executor })
    await token.transfer(voter2, amount, { from: executor })
    await token.transfer(voter3, amount, { from: executor })
    await token.transfer(voter4, amount, { from: executor })
    await token.transfer(voter5, amount, { from: executor })

    // Deploy timelock
    const minDelay = 1 // How long do we have to wait until we can execute after a passed proposal

    // In addition to passing minDelay, we also need to pass 2 arrays.
    // The 1st array contains addresses of those who are allowed to make a proposal.
    // The 2nd array contains addresses of those who are allowed to make executions.

    await deployer.deploy(Timelock, minDelay, [proposer], [executor])
    const timelock = await Timelock.deployed()
```

Deployment Script (2/2)

2_deploy_contracts.js

```
35  // Deploy governanace
36  const quorum = 5 // Percentage of total supply of tokens needed to aprove proposals (5%)
37  const votingDelay = 0 // How many blocks after proposal until voting becomes active
38  const votingPeriod = 5 // How many blocks to allow voters to vote
39
40  await deployer.deploy(Governance, token.address, timelock.address, quorum, votingDelay, votingPeriod)
41  const governance = await Governance.deployed()
42
43  // Deploy Treasury
44
45  // Timelock contract will be the owner of our treasury contract.
46  // In the provided example, once the proposal is successful and executed,
47  // timelock contract will be responsible for calling the function.
48
49  const funds = web3.utils.toWei('25', 'ether')
50
51  await deployer.deploy(Treasury, executor, { value: funds })
52  const treasury = await Treasury.deployed()
53
54  await treasury.transferOwnership(timelock.address, { from: executor })
55
56  // Assign roles
57
58  // You can view more information about timelock roles from the openzeppelin documentation:
59  // --> https://docs.openzeppelin.com/contracts/4.x/api/governance#timelock-proposer
60  // --> https://docs.openzeppelin.com/contracts/4.x/api/governance#timelock-executor
61
62  const proposerRole = await timelock.PROPOSER_ROLE()
63  const executorRole = await timelock.EXECUTOR_ROLE()
64
65  await timelock.grantRole(proposerRole, governance.address, { from: executor })
66  await timelock.grantRole(executorRole, governance.address, { from: executor })
67  };
```

The ownership of contract should be transferred to the timelock contract in order to be executed by the timelock after a proposal passed

Grant a proposer role and an executor role to the governance contract

Execution Script (1/4)

scripts/1_create_proposal.js

Prepare voters as delegators

```
Funds released? false
Funds inside of treasury: 25 ETH
```

```
1  const Token = artifacts.require("Token")
2  const Timelock = artifacts.require("Timelock")
3  const Governance = artifacts.require("Governance")
4  const Treasury = artifacts.require("Treasury")
5
6  module.exports = async function (callback) {
7      const [executor, proposer, voter1, voter2, voter3, voter4, voter5] = await web3.eth.getAccounts()
8
9      let isReleased, funds, blockNumber, proposalState, vote
10
11     const amount = web3.utils.toWei('5', 'ether')
12
13     // delegate voting to themselves
14     const token = await Token.deployed()
15     await token.delegate(voter1, { from: voter1 })
16     await token.delegate(voter2, { from: voter2 })
17     await token.delegate(voter3, { from: voter3 })
18     await token.delegate(voter4, { from: voter4 })
19     await token.delegate(voter5, { from: voter5 })
20
21     // get the status of Treasury
22     const treasury = await Treasury.deployed()
23
24     isReleased = await treasury.isReleased()
25     console.log(`Funds released? ${isReleased}`)
26
27     funds = await web3.eth.getBalance(treasury.address)
28     console.log(`Funds inside of treasury: ${web3.utils.fromWei(funds.toString(), 'ether')} ETH\n`)
```

Execution Script (2/4)

scripts/1_create_proposal.js

Propose phase

```
Created Proposal: 100758552249868773468698488155308693203920312959842149531265319478382688296537
```

```
Current state of proposal: 0 (Pending)
```

```
Proposal created on block 21
```

```
Proposal deadline on block 26
```

```
Current blocknumber: 21
```

```
Number of votes required to pass: 50
```

```
30    // Propose
31    const governance = await Governance.deployed()
32    const encodedFunction = await treasury.contract.methods.releaseFunds().encodeABI()
33    const description = "Release Funds from Treasury"
34
35    // propose
36    const tx = await governance.propose([treasury.address], [0], [encodedFunction], description, { from:
37
38    // get a proposal ID
39    const id = tx.logs[0].args.proposalId
40    console.log(`Created Proposal: ${id.toString()}\n`)
41
42    // get a current proposal state (Pending)
43    proposalState = await governance.state.call(id)
44    console.log(`Current state of proposal: ${proposalState.toString()} (Pending) \n`)
45
46    // get a block number where the proposal was created
47    const snapshot = await governance.proposalSnapshot.call(id)
48    console.log(`Proposal created on block ${snapshot.toString()}`)
49
50    // get a block number where the voting will end (snapshot + votingPeriod)
51    const deadline = await governance.proposalDeadline.call(id)
52    console.log(`Proposal deadline on block ${deadline.toString()}\n`)
53
54    // get a current block number
55    blockNumber = await web3.eth.getBlockNumber()
56    console.log(`Current blocknumber: ${blockNumber}\n`)
57
58    // get the required quorum
59    // we set it to 5% of total supply of tokens (5% of 1000 = 50 votes)
60    const quorum = await governance.quorum(blockNumber - 1)
61    console.log(`Number of votes required to pass: ${web3.utils.fromWei(quorum.toString(), 'ether')}\n`)
```

Execution Script (3/4)

scripts/1_create_proposal.js

Vote phase

```
Casting votes...

Current state of proposal: 1 (Active)
```

```
Votes For: 150
Votes Against: 50
Votes Neutral: 50
```

```
Current blocknumber: 27
```

```
Current state of proposal: 4 (Succeeded)
```

```
63    // Vote
64    console.log(`Casting votes...\n`)
65
66    // 0 = Against, 1 = For, 2 = Abstain
67    vote = await governance.castVote(id, 1, { from: voter1 })
68    vote = await governance.castVote(id, 1, { from: voter2 })
69    vote = await governance.castVote(id, 1, { from: voter3 })
70    vote = await governance.castVote(id, 0, { from: voter4 })
71    vote = await governance.castVote(id, 2, { from: voter5 })
72
73    // States: Pending, Active, Canceled, Defeated, Succeeded, Queued, Expired, Executed
74    proposalState = await governance.state.call(id)
75    console.log(`Current state of proposal: ${proposalState.toString()} (Active) \n`)
76
77    // NOTE: Transfer serves no purposes,
78    // it's just used to fast foward one block after the voting period ends
79    // because minDelay for queuing is set to 1 (block)
80    await token.transfer(proposer, amount, { from: executor })
81
82    // get voting results
83    const { againstVotes, forVotes, abstainVotes } = await governance.proposalVotes.call(id)
84    console.log(`Votes For: ${web3.utils.fromWei(forVotes.toString(), 'ether')}`)
85    console.log(`Votes Against: ${web3.utils.fromWei(againstVotes.toString(), 'ether')}`)
86    console.log(`Votes Neutral: ${web3.utils.fromWei(abstainVotes.toString(), 'ether')}\n`)
87
88    blockNumber = await web3.eth.getBlockNumber()
89    console.log(`Current blocknumber: ${blockNumber}\n`)
90
91    // check if the proposal passes
92    proposalState = await governance.state.call(id)
93    console.log(`Current state of proposal: ${proposalState.toString()} (Succeeded) \n`)
```

Execution Script (4/4)

scripts/1_create_proposal.js

Queue & Execute phase

```
Current state of proposal: 5 (Queued)
```

```
Current state of proposal: 7 (Executed)
```

```
Funds released? true
Funds inside of treasury: 0 ETH
```

```
95    // Queue
96    const hash = web3.utils.sha3("Release Funds from Treasury")
97    await governance.queue([treasury.address], [0], [encodedFunction], hash, { from: executor })
98
99    proposalState = await governance.state.call(id)
100   console.log(`Current state of proposal: ${proposalState.toString()} (Queued) \n`)
101
102   // Execute
103   await governance.execute([treasury.address], [0], [encodedFunction], hash, { from: executor })
104
105   proposalState = await governance.state.call(id)
106   console.log(`Current state of proposal: ${proposalState.toString()} (Executed) \n`)
107
108   isReleased = await treasury.isReleased()
109   console.log(`Funds released? ${isReleased}`)
110
111   funds = await web3.eth.getBalance(treasury.address)
112   console.log(`Funds inside of treasury: ${web3.utils.fromWei(funds.toString(), 'ether')} ETH\n`)
113
114   callback()
115 }
```

Behind OpenZeppelin Governance

Governor Architecture



Defining voting options (3 options) & counting votes (_countVote(), _quorumReached(), _voteSucceeded())

Setting parameters votingDelay, votingPeriod, proposalThreshold

Binding the execution to TimelockController enforcing the minDelay before the execution
Overriding Governor's functions such as state(), _execute(), _cancel()

Wrapper contract for tokens (ERC20, ERC721)
Binding the token to ERC20Votes
Getting the voting weight (_getVotes())

Setting up the roles

TimelockController.sol

Set up the admin role to this contract

execute() function has the modifier to check if the caller has a role of executor or the executor role open to anyone (address(0))

```
84    bytes32 public constant TIMELOCK_ADMIN_ROLE = keccak256("TIMELOCK_ADMIN_ROLE");
85    bytes32 public constant PROPOSER_ROLE = keccak256("PROPOSER_ROLE");
86    bytes32 public constant EXECUTOR_ROLE = keccak256("EXECUTOR_ROLE");
87    bytes32 public constant CANCELLER_ROLE = keccak256("CANCELLER_ROLE");
88
89    constructor(uint256 minDelay, address[] memory proposers,
90        address[] memory executors, address admin) {
91        _setRoleAdmin(TIMELOCK_ADMIN_ROLE, TIMELOCK_ADMIN_ROLE);
92        _setRoleAdmin(PROPOSER_ROLE, TIMELOCK_ADMIN_ROLE);
93        _setRoleAdmin(EXECUTOR_ROLE, TIMELOCK_ADMIN_ROLE);
94        _setRoleAdmin(CANCELLER_ROLE, TIMELOCK_ADMIN_ROLE);
95
96        // self administration
97        _setupRole(TIMELOCK_ADMIN_ROLE, address(this));
98
99        // optional admin
100       if (admin != address(0)) {
101           _setupRole(TIMELOCK_ADMIN_ROLE, admin);
102       }
103
104       // register proposers and cancellers
105       for (uint256 i = 0; i < proposers.length; ++i) {
106           _setupRole(PROPOSER_ROLE, proposers[i]);
107           _setupRole(CANCELLER_ROLE, proposers[i]);
108       }
109
110       // register executors
111       for (uint256 i = 0; i < executors.length; ++i) {
112           _setupRole(EXECUTOR_ROLE, executors[i]);
113       }
114
115       _minDelay = minDelay;
116       emit MinDelayChange(0, minDelay);
117   }
118
119   modifier onlyRoleOrOpenRole(bytes32 role) {
120       if (!hasRole(role, address(0))) {
121           _checkRole(role, _msgSender());
122       }
123       _;
124   }
```

Proposing

Governor.sol

Check if the proposer has more voting power than the proposal threshold

The proposal ID is the hash value of all input of the proposal

Setting the voting start time and end time

Adding a proposal to the proposals variable
Note. it doesn't store the proposal content to save storage

```
268     function propose(
269         address[] memory targets,
270         uint256[] memory values,
271         bytes[] memory calldatas,
272         string memory description
273     ) public virtual override returns (uint256) {
274         address proposer = _msgSender();
275         uint256 currentTimepoint = clock();
276
277         require(
278             getVotes(proposer, currentTimepoint - 1) >= proposalThreshold(),
279             "Governor: proposer votes below proposal threshold"
280         );
281
282         uint256 proposalId = hashProposal(targets, values, calldatas, keccak256(bytes(description)));
283
284         require(targets.length == values.length, "Governor: invalid proposal length");
285         require(targets.length == calldatas.length, "Governor: invalid proposal length");
286         require(targets.length > 0, "Governor: empty proposal");
287         require(_proposals[proposalId].voteStart == 0, "Governor: proposal already exists");
288
289         uint256 snapshot = currentTimepoint + votingDelay();
290         uint256 deadline = snapshot + votingPeriod();
291
292         _proposals[proposalId] = ProposalCore({
293             proposer: proposer,
294             voteStart: SafeCast.toUint64(snapshot),
295             voteEnd: SafeCast.toUint64(deadline),
296             executed: false,
297             canceled: false,
298             __gap_unused0: 0,
299             __gap_unused1: 0
300         });
```

Casting a vote

Governor.sol

```
564     function _castVote(
565         uint256 proposalId,
566         address account,
567         uint8 support,
568         string memory reason,
569         bytes memory params
570     ) internal virtual returns (uint256) {
571         ProposalCore storage proposal = _proposals[proposalId];
572         require(state(proposalId) == ProposalState.Active, "Governor: vote not currently active");
573
574         uint256 weight = _getVotes(account, proposal.voteStart, params);
575         _countVote(proposalId, account, support, weight, params);
576
577         if (params.length == 0) {
578             emit VoteCast(account, proposalId, support, weight, reason);
579         } else {
580             emit VoteCastWithParams(account, proposalId, support, weight, reason, params);
581         }
582
583         return weight;
584     }
```

GovernorVotes.sol

```
48      function _getVotes(
49          address account,
50          uint256 timepoint,
51          bytes memory /*params*/
52      ) internal view virtual override returns (uint256) {
53          return token.getPastVotes(account, timepoint);
54      }
55  }
```

ERC20Votes.sol

```
95      function getPastVotes(address account, uint256 timepoint) public view virtual
96          require(timepoint < clock(), "ERC20Votes: future lookup");
97          return _checkpointsLookup(_checkpoints[account], timepoint);
98      }
```

GovernorCountingSimple.sol

```
78      function _countVote(
79          uint256 proposalId,
80          address account,
81          uint8 support,
82          uint256 weight,
83          bytes memory // params
84      ) internal virtual override {
85          ProposalVote storage proposalVote = _proposalVotes[proposalId];
86
87          require(!proposalVote.hasVoted[account], "GovernorVotingSimple: vote already cast");
88          proposalVote.hasVoted[account] = true;
89
90          if (support == uint8(VoteType.Against)) {
91              proposalVote.againstVotes += weight;
92          } else if (support == uint8(VoteType.For)) {
93              proposalVote.forVotes += weight;
94          } else if (support == uint8(VoteType.Abstain)) {
95              proposalVote.abstainVotes += weight;
96          } else {
97              revert("GovernorVotingSimple: invalid value for enum VoteType");
98          }
99      }
100 }
```

Checking the vote succeeded

Governor.sol

```
163  function state(uint256 proposalId) public view virtual override returns (ProposalState) {
164      ProposalCore storage proposal = _proposals[proposalId];
165
166      if (proposal.executed) {
167          return ProposalState.Executed;
168      }
169
170      if (proposal.canceled) {
171          return ProposalState.Canceled;
172      }
173
174      uint256 snapshot = proposalSnapshot(proposalId);
175
176      if (snapshot == 0) {
177          revert("Governor: unknown proposal id");
178      }
179
180      uint256 currentTimepoint = clock();
181
182      if (snapshot >= currentTimepoint) {
183          return ProposalState.Pending;
184      }
185
186      uint256 deadline = proposalDeadline(proposalId);
187
188      if (deadline >= currentTimepoint) {
189          return ProposalState.Active;
190      }
191
192      if (_quorumReached(proposalId) && _voteSucceeded(proposalId)) {
193          return ProposalState.Succeeded;
194      } else {
195          return ProposalState.Defeated;
196      }
197  }
```

GovernorCountingSimple.sol

```
60      function _quorumReached(uint256 proposalId) internal view virtual override returns (bool) {
61          ProposalVote storage proposalVote = _proposalVotes[proposalId];
62
63          return quorum(proposalSnapshot(proposalId)) <= proposalVote.forVotes + proposalVote.abstainVotes;
64      }
65
66      /**
67       * @dev See {Governor-_voteSucceeded}. In this module, the forVotes must be strictly over the againstVotes.
68       */
69      function _voteSucceeded(uint256 proposalId) internal view virtual override returns (bool) {
70          ProposalVote storage proposalVote = _proposalVotes[proposalId];
71
72          return proposalVote.forVotes > proposalVote.againstVotes;
73      }
```

GovernorVotesQuorumFraction.sol

```
71      /**
72       * @dev Returns the quorum for a timepoint, in terms of number of votes: `supply * numerator / denominator`.
73       */
74      function quorum(uint256 timepoint) public view virtual override returns (uint256) {
75          return (token.getPastTotalSupply(timepoint) * quorumNumerator(timepoint)) / quorumDenominator();
76      }
```

ERC20Votes.sol

```
108     function getPastTotalSupply(uint256 timepoint) public view virtual override returns (uint256) {
109         require(timepoint < clock(), "ERC20Votes: future lookup");
110         return _checkpointsLookup(_totalSupplyCheckpoints, timepoint);
111     }
```

Queuing

GovernorTimelockControl.sol

Check the state of a proposal with the proposal ID

By the proposal ID, the proposal can be verified without saving the proposal content

```
90    function queue(
91        address[] memory targets,
92        uint256[] memory values,
93        bytes[] memory calldatas,
94        bytes32 descriptionHash
95    ) public virtual override returns (uint256) {
96        uint256 proposalId = hashProposal(targets, values, calldatas, descriptionHash);
97
98        require(state(proposalId) == ProposalState.Succeeded, "Governor: proposal not successful");
99
100       uint256 delay = _timelock.getMinDelay();
101       _timelockIds[proposalId] = _timelock.hashOperationBatch(targets, values, calldatas, 0, descrip
102       _timelock.scheduleBatch(targets, values, calldatas, 0, descriptionHash, delay);
103
104       emit ProposalQueued(proposalId, block.timestamp + delay);
105
106       return proposalId;
107   }
```

TimelockController.sol

Schedule a proposal to execute after minDelay

Set the timestamp when the proposal is able to be executed after

```
244   function scheduleBatch(
245       address[] calldata targets,
246       uint256[] calldata values,
247       bytes[] calldata payloads,
248       bytes32 predecessor,
249       bytes32 salt,
250       uint256 delay
251   ) public virtual onlyRole(PROPOSER_ROLE) {
252       require(targets.length == values.length, "TimelockController: length mismatch");
253       require(targets.length == payloads.length, "TimelockController: length mismatch");
254
255       bytes32 id = hashOperationBatch(targets, values, payloads, predecessor, salt);
256       _schedule(id, delay);
257       for (uint256 i = 0; i < targets.length; ++i) {
258           emit CallScheduled(id, i, targets[i], values[i], payloads[i], predecessor, delay);
259       }
260       if (salt != bytes32(0)) {
261           emit CallSalt(id, salt);
262       }
263   }
264
265   function _schedule(bytes32 id, uint256 delay) private {
266       require(!isOperation(id), "TimelockController: operation already scheduled");
267       require(delay >= getMinDelay(), "TimelockController: insufficient delay");
268       _timestamps[id] = block.timestamp + delay;
269   }
```

Checking the state of a proposal

TimelockController.sol

if _timestamp[id] is
0: unset (Vote)
>1: queued (Queue)
1: done (Execute)

Note.
queue() and execute() function
will not be called automatically
even after the condition is satisfied.

Someone should call the functions explicitly
and then check if they can be executed

```
7     uint256 internal constant _DONE_TIMESTAMP = uint256(1);
8     mapping(bytes32 => uint256) private _timestamps;
9
10    // check if it is queued or done (already scheduled)
11    function isOperation(bytes32 id) public view virtual returns (bool) {
12        return getTimestamp(id) > 0;
13    }
14
15    // check if is is queued (the timestamp was already set)
16    function isOperationPending(bytes32 id) public view virtual returns (bool) {
17        return getTimestamp(id) > _DONE_TIMESTAMP;
18    }
19
20    // check if the execution is ready since minDelay has passed
21    function isOperationReady(bytes32 id) public view virtual returns (bool) {
22        uint256 timestamp = getTimestamp(id);
23        return timestamp > _DONE_TIMESTAMP && timestamp <= block.timestamp;
24    }
25
26    function isOperationDone(bytes32 id) public view virtual returns (bool) {
27        return getTimestamp(id) == _DONE_TIMESTAMP;
28    }
29
30    // 0 for unset operations, 1 for done operations, timestamp for queued ops
31    function getTimestamp(bytes32 id) public view virtual returns (uint256) {
32        return _timestamps[id];
33    }
```

Executing

TimelockController.sol

Only the executor role can run the function

Only the executor role can run the function

Check if the execution is ready

Send a tx request to the target contract with the ETH value and calldata

Check if minDelay has passed since the voting ended

```
327     function executeBatch(
328         address[] calldata targets,
329         uint256[] calldata values,
330         bytes[] calldata payloads,
331         bytes32 predecessor,
332         bytes32 salt
333     ) public payable virtual onlyRoleOrOpenRole(EXECUTOR_ROLE) {
334         require(targets.length == values.length, "TimelockController: length mismatch");
335         require(targets.length == payloads.length, "TimelockController: length mismatch");
336
337         bytes32 id = hashOperationBatch(targets, values, payloads, predecessor, salt);
338
339         _beforeCall(id, predecessor);
340         for (uint256 i = 0; i < targets.length; ++i) {
341             address target = targets[i];
342             uint256 value = values[i];
343             bytes calldata payload = payloads[i];
344             _execute(target, value, payload);
345             emit CallExecuted(id, i, target, value, payload);
346         }
347         _afterCall(id);
348     }
349
350     function _execute(address target, uint256 value, bytes calldata data) internal virtual {
351         (bool success, ) = target.call{value: value}(data);
352         require(success, "TimelockController: underlying transaction reverted");
353     }
354
355     function _beforeCall(bytes32 id, bytes32 predecessor) private view {
356         require(isOperationReady(id), "TimelockController: operation is not ready");
357         require(predecessor == bytes32(0) || isOperationDone(predecessor), "TimelockController:
358     }
```

Best practices to learn

- Create a main contract (Governor) and make it customizable through contract inheritance and wrapper contracts.

- Set up roles to control access to functions and set up states to control that only functions that fit the state are executed.

- To save blockchain storage, don't store large input parameters(proposal content). Instead, store only the hash value(proposalId) of the parameters. By hashing them in function calls and checking if it is the same value as before, we can verify if the parameter is the same in a series of function calls. The large parameters may be stored offchain.

Upgradeable Contracts

A smart contract is **immutable**.

Can we **upgrade** a smart contract?

Three options of upgrading

- Set upgradable parameters
  e.g., setMiningRewards()

- Migrate all states and users from the old contract to the new one
  e.g., Uniswap V1, V2, V3 (Migrate from V2 to V3)

- Use a proxy

How a proxy works to upgrade a contract



https://www.youtube.com/watch?v=JgSj7IiE4jA

DelegateCall

- **DelegateCall** is a low-level Solidity opcode that allows a contract to execute code from another contract, but it **using the state and the storage of the calling contract**.

- The syntax for DelegateCall

  (bool success, bytes memory returnData) = address.delegatecall(bytes memory data);

  the address is the address of contract to execute, and the data is the encoded function call to execute

- Call vs. DelegateCall





https://https://www.linkedin.com/pulse/delegatecall-solidity-some-code-examples-johnny-time/

DelegateCall Example

DelegateCallExample.sol

```sol
// SPDX-License-Identifier: MIT
pragma solidity ^0.8.9;

contract TargetContract {
    uint public a;
    function setA(uint _a) public {
        a = _a;
    }
}

contract CallerContract {
    address public aAddress;
    uint public b;

    constructor(address _aAddress) {
        aAddress = _aAddress;
    }

    function setA(uint _a) public {
        (bool success, ) = aAddress.delegatecall(
            abi.encodeWithSignature("setA(uint256)", _a)
        );
        require(success, "delegatecall failed");
    }

    function setB(uint _b) public {
        b = _b;
    }
}
```

variable a will be stored on CallerContract

Proxy.sol

```sol
// SPDX-License-Identifier: MIT
pragma solidity ^0.8.9;

contract Proxy {
    address private _implementation;

    function setImplementation(address implementation) external {
        _implementation = implementation;
    }

    fallback() external payable {
        address impl = _implementation;
        assembly {
            calldatacopy(0, 0, calldatasize())
            let result := delegatecall(gas(), impl, 0, calldatasize(), 0, 0)
            returndatacopy(0, 0, returndatasize())
            switch result
            case 0 {
                revert(0, returndatasize())
            }
            default {
                return(0, returndatasize())
            }
        }
    }
```

https://https://www.linkedin.com/pulse/delegatecall-solidity-some-code-examples-johnny-time/

Implementing Upgradeable contracts with OpenZeppelin & Hardhat

Implementation contracts



https://www.youtube.com/watch?v=JgSj7IiE4jA
https://github.com/t4sk/hello-oz-upgradeable

Deploy & Upgrade contracts

scripts/deploy_box_v1.js

```
1  const { ethers, upgrades } = require("hardhat");
2
3  async function main() {
4      const Box = await ethers.getContractFactory("Box");
5      console.log("Deploying Box...");
6      const box = await upgrades.deployProxy(Box, [42], {
7          initializer: "initialize",
8      });
9      await box.deployed();
10     console.log("Box deployed to:", box.address);
11 }
12
13 main();
```

TransparentUpgradeableProxy



ProxyAdmin

Box

Etherscan: https://sepolia.etherscan.io/address/0x20efda938e7c1bf25ba7dc6b7a4ac8075b7dfbda

scripts/upgrade_box_v2.js

```
1  const { ethers, upgrades } = require("hardhat");
2
3  const PROXY = "0x2C384EE352EEa99Fc8B6dDC7e6b5664397CED172";
4
5  async function main() {
6      const BoxV2 = await ethers.getContractFactory("BoxV2");
7      console.log("Upgrading Box...");
8      await upgrades.upgradeProxy(PROXY, BoxV2);
9      console.log("Box upgraded");
10 }
11
12 main();
```

ProxyAdmin.upgrade()



BoxV2

Etherscan: https://sepolia.etherscan.io/address/0x20efda938e7c1bf25ba7dc6b7a4ac8075b7dfbda

Run contracts with etherscan

TransparentUpgradeableProxy

https://sepolia.etherscan.io/address/0x2c384ee352eea99fc8b6ddc7e6b5664397ced172#code







hardhat config file and commands

hardhat.config.js

```
require("@openzeppelin/hardhat-upgrades");
require("@nomicfoundation/hardhat-toolbox");

const dotenv = require('dotenv');
dotenv.config();

module.exports = {
  solidity: "0.8.10",
  networks: {
    sepolia: {
      url: `https://eth-sepolia.g.alchemy.com/v2/${process.env.ALCHEMY_API_KEY}`,
      accounts: [process.env.SEPOLIA_PRIVATE_KEY]
    }
  },
  etherscan: {
    apiKey: process.env.ETHERSCAN_API_KEY,
  },
};
```

commands

```
// install hardhat
npm install --save-dev hardhat
npx hardhat
<Create an empty hardhat.config.js>

// install packages
npm install --save-dev @nomicfoundation/hardhat-toolbox
npm install @openzeppelin/contracts
npm install @openzeppelin/contracts-upgradeable
npm install @openzeppelin/hardhat-upgrades
npm install dotenv
<Create and set up .env file>

// compile and deploy
npx hardhat compile

// deploy the Box contract
npx hardhat run --network sepolia .\scripts\deploy_box_v1.js
// verify the Box code to etherscan
npx hardhat verify --network sepolia 0x8E6F7b6efffb21aA320909E90d5613ac7fe796F4

// deploy the BoxV2 contract
npx hardhat run --network sepolia .\scripts\upgrade_box_v2.js
// verify the BoxV2 code to etherscan
npx hardhat verify --network sepolia 0x296eDded27E9c081762a1627DDDEb999F09cD18e
```

Upgradeable ERC20 tokens



https://www.youtube.com/watch?v=Vt20jCu8OC8
https://github.com/t4sk/hello-oz-upgradeable

Restrictions of upgradeable contracts

- **Use initialize(), a regular function to run all the setup logic instead of a constructor**
- Inherit an Initializable contract and use an initializer modifier in order not to call initialize() multiple times
- Avoid initial values in field declarations. Make sure that all initial values are set in an initializer function (ok to define constant state variables)
- Invoke the _disableInitializers function in the constructor to automatically lock it when it is deployed

- Use @openzeppelin/contracts-upgradeable for libraries and contracts instead of @openzeppelin/contracts

- You cannot change the order in which the contract state variables are declared, nor their type. **If you need to introduce a new variable, make sure you always do so at the end.**

- Read OpenZeppelin docs for more restrictions

Wrap-up

We Learned

- Access Control
- Onchain governance
- Developing onchain governance with OpenZeppelin
- Technical detail on OpenZeppelin governance
- Upgradeable contract
- Developing upgradeable contract with OpenZeppelin

Revisit: Web3 Stack from the first lecture



In terms of web3 technology, we covered three layers from the bottom. You had a skill to develop your own basic web3 apps

Lecture 19
Access control, on-chain governance, upgradeable contracts

Lecture 15, 17
non-standard token, ERC20, ERC721(NFT), ERC1155, ERC5192(SBT), Oracle, IPFS

Lecture 1 – 12
Web3 staaks, blockchain tech, programming solidity, smart contracts, building dapps, web3 security